
Estimation results

Example (Simulated data)
. describe

Contains data from etable.dta
Observations: 200

Variables: 6 20 Sep 2021 08:32

Variable Storage Display Value
name type format label Variable label

x1 float %9.0g X_1
x2 float %9.0g X_2
f1 byte %9.0g f1 F_1
f2 byte %9.0g f2 F_2
y1 byte %9.0g Binary outcome
y2 float %9.0g Continuous outcome

Sorted by:



Estimation results

Estimation commands
Fit models and post their results to e().

ereturn list

I Scalars
I e(N) – sample size
I e(ll) – log likelihood

I Macros
I e(cmdline) – command as typed
I e(depvar) – dependent variable
I e(title) – estimation title

I Matrices
I e(b) – fitted values
I e(V) – estimated variance for e(b)



Estimation results

Coefficient table
Show fitted values, standard errors, tests, and confidence
intervals.
matrix list r(table)

Columns conform with e(b).
Rows are the displayed values:

I b – fitted values (β̂)
I se – standard error of b
I z or t – test for H0 : β = 0
I pvalue – p-value for H0

I ll – lower confidence limit
I ul – lower confidence limit
I df – degrees of freedom



Estimation results

System variables
Provide easy access to elements of e(b) and e(V).

I _b – elements of e(b)
I _se – square root of diagonal elements of e(V)

Use _b with postestimation commands lincom, nlcom, test,
and testnl.

Example

. use etable

. logit y1 x1 x2 i.f1

. logit, coeflegend

. lincom _b[y1:x1] - 2*_b[y1:x2]

. nlcom _b[y1:x1] / _b[y1:x2]



Estimation results

System variables new in Stata 17
Provide easy access to elements of r(table).

I _r_b – b elements of r(table)
I _r_se – se elements of r(table)
I _r_z – z or t elements of r(table)
I _r_p – pvalue elements of r(table)
I _r_lb – ll elements of r(table)
I _r_ub – ul elements of r(table)
I _r_df – df elements of r(table)

In a future update

I _r_z_abs – absolute value of _r_z



Estimation results

System variables, continued

I r(table) is fleeting compared to the results in e().
I _r_b and friends pull from a hidden copy stored with e().

I estimates store
I estimates restore
I estimates save
I estimates use



Estimation results

Replay estimation results
_r_b and friends automatically update.

Example
. quietly logit

. display _b[x1]
1.322184

. display _r_b[x1]
1.322184

. quietly logit, or

. display _b[x1]
1.322184

. display _r_b[x1]
3.7516061



Estimation results

Estimation tables
Commands available before Stata 17:

I Stata
I estimates table
I putexcel
I putdocx table
I putpdf table
I dyndoc, dyntext, and markdown

I Community contributed
I outreg2 by R. Wada
I outreg and frmttable by J. L. Gallup
I estout and esttab by B. Jann
I asdoc by A. Shah
I markdoc by E. F. Haghish
I ...



Collections

New in Stata 17
collect is a suite of commands for building custom tables in
Stata.

Basic workflow
I collect get – consume results into a collection

I collect
[
get

]
: – prefix

I collect layout – arrange the results in your table
I collect style – control appearance and behaviors
I collect query – show collection properties
I collect label – manipulate and show labels
I collect export – your table to a document



Collections

collect get
Specify what results to consume.

Example
. quietly logit y1 x1 x2 i.f1
. collect get e()
. collect dir

Collections in memory
Current: default

Name No. items

default 72



Collections

collect: prefix
Will get results from e() if they were created by running the
prefixed command.

Example

. collect : logit y1 x1 x2 i.f1



Collections

Items and tags
collect get and collect: consume results by adding
them as items to a collection.

I An item is a single number or string.
I Items are tagged for reference in style edits and the layout.
I Tags are composed from dimension-level pairs.



Collections

Scalar results
An item from e(N) has tag elements:

I result[N]

I result_type[scalar]

I program_class[eclass]

Macro results
An item from e(cmd) has tag elements:

I result[cmd]

I result_type[macro]

I program_class[eclass]



Collections

Matrix results
Also have tag elements for row and column names.

Example
. matrix list e(rules)

e(rules)[1,4]
c1 c2 c3 c4

r1 0 0 0 0

An item from e(rules)["r1","c1"] has tag elements:
I result[rules]

I result_type[matrix]

I program_class[eclass]

I rowname[r1]

I colname[c1]



Collections

Special matrices
e(b) and e(V) are not consumed like other matrices.

r(table) is consumed using the new system variables.

r(table)["b","y1:x1"] is _r_b[y1:x1]

An item from this result has tag elements:

I result[_r_b]

I result_type[matrix]

I program_class[eclass]

I coleq[y1]

I colname[x1]



Collections

Example (list the dimensions)
. collect dims

Collection dimensions
Collection: default

Dimension No. levels

Layout, style, header, label
cmdset 1
coleq 1

colname 9
colname_remainder 1

f1 2
program_class 1

result 44
result_type 3

rowname 1

Style only
border_block 4

cell_type 4



Collections

cmdset
Index for each set of results in the collection.

Example (list the levels/labels for cmdset)
. collect levelsof cmdset

Collection: default
Dimension: cmdset

Levels: 1

. collect label list cmdset, all

Collection: default
Dimension: cmdset

Label: Command results index
Level labels:

1



Collections

Example (add label to describe this results set)
. collect label values cmdset 1 "log(Odds ratio)"

. collect label list cmdset

Collection: default
Dimension: cmdset

Label: Command results index
Level labels:

1 log(Odds ratio)



Collections

coleq
Column equations taken from matrix results.

Look like a variable? Get the variable’s label.

Example
. collect levelsof coleq

Collection: default
Dimension: coleq

Levels: y1

. collect label list coleq

Collection: default
Dimension: coleq

Label: Depvars, parameters, and column equations
Level labels:

y1 Binary outcome



Collections

colname
Column names taken from matrix results.

Look like a variable? Get the variable’s label.

Example
. collect levelsof colname

Collection: default
Dimension: colname

Levels: x1 x2 0.f1 1.f1 c1 c2 c3 c4 _cons

. collect label list colname

Collection: default
Dimension: colname

Label: Covariate names and column names
Level labels:

_cons Intercept
f1 F_1
x1 X_1
x2 X_2



Collections

Factor variables
Become dimensions, if found in colname.

Their labels are also consumed.

Example
. collect levelsof f1

Collection: default
Dimension: f1

Levels: 0 1

. collect label list f1

Collection: default
Dimension: f1

Label: F_1
Level labels:

0 Off
1 On



Collections

result
Filled with the names of results consumed from e(), and
includes the r(table) system variables.

. collect levelsof result
(output too long, omitted)

collect has command-specific labels for many results.

. collect label list result, all
(output too long, omitted)

_r_ci is a special composite level in result, it’s items are
composed from items tagged with _r_lb and _r_ub.



Collections

colname_remainder
Contains the remaining column stripe pieces when factor
variables are removed.

Possibly useful for arranging/selecting interaction elements in
the layout.

rowname
Similar to colname.

border_block and cell_type
Are not part of tags.

Used for style targeting.



Layout and appearance

collect layout
Use dimensions to select and arrange items into a table.

Without arguments, show the current layout information and
possibly show you a table.

Example (default layout is empty)
. collect layout

Your layout specification does not identify any items.



Layout and appearance

Syntax

collect layout (rowspec) (colspec)

where rowspec and colspec are composed from dimensions
and their interactions (#).

Coefficient table layout
rowspec

coleq#colname

colspec

result[_r_b _r_se _r_z _r_p _r_ci]



Layout and appearance

Example (coefficient table layout)
. collect layout (coleq#colname) (result[_r_b _r_se _r_z _r_p _r_ci])

Collection: default
Rows: coleq#colname

Columns: result[_r_b _r_se _r_z _r_p _r_ci]
Table 1: 6 x 5

Coefficient Std. error z p-value 95% CI

Binary outcome
X_1 1.322184 .5985862 2.21 0.027 .1489766 2.495391
X_2 -1.064822 .5836827 -1.82 0.068 -2.208819 .079175
Off 0 0
On 2.049094 .3342964 6.13 0.000 1.393885 2.704303
Intercept -.7792675 .4344927 -1.79 0.073 -1.630858 .0723225



Layout and appearance

collect style header
Controls the appearance of dimensions and their levels in row
and column headers.

Example (hide the equation)
. collect query header coleq

Dimension header styles
Collection: default
Dimension: coleq

Level:
Show title: hide
Show level: label

. collect style header coleq, level(hide)



Layout and appearance

collect preview
Shows a preview of your table.

Example (bye bye equation)
. collect preview

Coefficient Std. error z p-value 95% CI

X_1 1.322184 .5985862 2.21 0.027 .1489766 2.495391
X_2 -1.064822 .5836827 -1.82 0.068 -2.208819 .079175
Off 0 0
On 2.049094 .3342964 6.13 0.000 1.393885 2.704303
Intercept -.7792675 .4344927 -1.79 0.073 -1.630858 .0723225



Layout and appearance

Example (hide base levels of factor variables)
. collect query showbase

Display base levels
Collection: default
Show base: all

. collect style showbase off

. collect preview

Coefficient Std. error z p-value 95% CI

X_1 1.322184 .5985862 2.21 0.027 .1489766 2.495391
X_2 -1.064822 .5836827 -1.82 0.068 -2.208819 .079175
On 2.049094 .3342964 6.13 0.000 1.393885 2.704303
Intercept -.7792675 .4344927 -1.79 0.073 -1.630858 .0723225



Layout and appearance

Example (show factor variable f1)
. collect query header f1

Dimension header styles
Collection: default
Dimension: f1

Level:
Show title: hide
Show level: label

. collect style header f1, title(label)

. collect preview

Coefficient Std. error z p-value 95% CI

X_1 1.322184 .5985862 2.21 0.027 .1489766 2.495391
X_2 -1.064822 .5836827 -1.82 0.068 -2.208819 .079175
F_1=On 2.049094 .3342964 6.13 0.000 1.393885 2.704303
Intercept -.7792675 .4344927 -1.79 0.073 -1.630858 .0723225



Layout and appearance

collect style row
Controls appearance and arrangement of row headers.

Example (remove binder on factor variables)
. collect query row
(output too long, omitted)

. collect style row stack, nobinder

. collect preview

Coefficient Std. error z p-value 95% CI

X_1 1.322184 .5985862 2.21 0.027 .1489766 2.495391
X_2 -1.064822 .5836827 -1.82 0.068 -2.208819 .079175
F_1
On 2.049094 .3342964 6.13 0.000 1.393885 2.704303

Intercept -.7792675 .4344927 -1.79 0.073 -1.630858 .0723225



Layout and appearance

collect style cell
Control appearance of cell contents in the table.

Example (fixed format for coefficient estimates)
. collect style cell result[_r_b _r_se _r_ci], nformat(%7.4f)

. collect preview

Coefficient Std. error z p-value 95% CI

X_1 1.3222 0.5986 2.21 0.027 0.1490 2.4954
X_2 -1.0648 0.5837 -1.82 0.068 -2.2088 0.0792
F_1
On 2.0491 0.3343 6.13 0.000 1.3939 2.7043

Intercept -0.7793 0.4345 -1.79 0.073 -1.6309 0.0723



Layout and appearance

Example (estimates table layout)
. collect layout (coleq#colname#result[_r_b _r_se]) (cmdset)

Collection: default
Rows: coleq#colname#result[_r_b _r_se]

Columns: cmdset
Table 1: 13 x 1

log(Odds ratio)

X_1
Coefficient 1.3222
Std. error 0.5986

X_2
Coefficient -1.0648
Std. error 0.5837

F_1
On
Coefficient 2.0491
Std. error 0.3343

Intercept
Coefficient -0.7793
Std. error 0.4345



Layout and appearance

Example (add parenthesis around standard errors)
. collect style cell result[_r_se], sformat("(%s)")

. collect preview

log(Odds ratio)

X_1
Coefficient 1.3222
Std. error (0.5986)

X_2
Coefficient -1.0648
Std. error (0.5837)

F_1
On
Coefficient 2.0491
Std. error (0.3343)

Intercept
Coefficient -0.7793
Std. error (0.4345)



Layout and appearance

Example (hide result levels)
. collect query header result

Dimension header styles
Collection: default
Dimension: result

Level:
Show title: hide
Show level: label

. collect style header result , level(hide)

. collect preview

log(Odds ratio)

X_1 1.3222
(0.5986)

X_2 -1.0648
(0.5837)

F_1
On 2.0491

(0.3343)
Intercept -0.7793

(0.4345)



Stars

collect stars
Label significant results.

Example (commonly used labeling rules)
. collect stars _r_p .01 "**" .05 "*"

Example (where are the stars???)
. collect preview

log(Odds ratio)

X_1 1.3222
(0.5986)

X_2 -1.0648
(0.5837)

F_1
On 2.0491

(0.3343)
Intercept -0.7793

(0.4345)



Stars

collect stars
Adds items tagged with result[stars] by default.

Example (update the layout)
. collect layout (coleq#colname#result[_r_b _r_se stars]) (cmdset)

Collection: default
Rows: coleq#colname#result[_r_b _r_se stars]

Columns: cmdset
Table 1: 11 x 1

log(Odds ratio)

X_1 1.3222
(0.5986)

*
X_2 -1.0648

(0.5837)
F_1
On 2.0491

(0.3343)

**
Intercept -0.7793

(0.4345)



Stars

collect stars
Can attach labels to items tagged with a different result.

Example (attach stars to the fitted values)
. collect stars _r_p .01 "**" .05 "*" , attach(_r_b)

. collect layout (coleq#colname#result[_r_b _r_se]) (cmdset)

Collection: default
Rows: coleq#colname#result[_r_b _r_se]

Columns: cmdset
Table 1: 9 x 1

log(Odds ratio)

X_1 1.3222*
(0.5986)

X_2 -1.0648
(0.5837)

F_1
On 2.0491**

(0.3343)
Intercept -0.7793

(0.4345)



Stars

Future update to Stata 17

I collect stars information will be stored in the style.
I collect will apply the stars labeling rules to

subsequently collected results.
I New option dimension to add stars as a dimension

instead of a level of result.



Stars
Example
. collect stars, dimension

. collect dims

Collection dimensions
Collection: default

Dimension No. levels

Layout, style, header, label
cmdset 1
coleq 1

colname 9
colname_remainder 1

f1 2
program_class 1

result 45
result_type 3

rowname 1
stars 2

Style only
border_block 4

cell_type 4

. collect levelsof stars

Collection: default
Dimension: stars

Levels: label value



Stars

Example (add stars dimension to the layout)
. collect layout (coleq#colname#result[_r_b _r_se]) (cmdset#stars)

Collection: default
Rows: coleq#colname#result[_r_b _r_se]

Columns: cmdset#stars
Table 1: 9 x 2

log(Odds ratio) log(Odds ratio)

X_1 1.3222 *
(0.5986)

X_2 -1.0648
(0.5837)

F_1
On 2.0491 **

(0.3343)
Intercept -0.7793

(0.4345)



Stars

collect style column
Controls appearance and arrangement of column headers.

Example (center duplicate column headers)
. collect query column

Column header styles
Collection: default

No delimiter: on
Delimiter: " # "

At delimiter: " @ "
Bar delimiter: " | "

Binder: "="
Duplicates: repeat
Position: top

Extra space: 0
Width: asis

. collect style column, dups(center)



Stars

Example
. collect preview

log(Odds ratio)

X_1 1.3222 *
(0.5986)

X_2 -1.0648
(0.5837)

F_1
On 2.0491 **

(0.3343)
Intercept -0.7793

(0.4345)



Stars

Example (left align the stars labels)
. collect style cell stars[label], halign(left)

. collect preview

log(Odds ratio)

X_1 1.3222 *
(0.5986)

X_2 -1.0648
(0.5837)

F_1
On 2.0491 **

(0.3343)
Intercept -0.7793

(0.4345)



Stars

Example (add odds ratios results)
. quietly logit, or

. collect get e()

. collect label levels cmdset 2 "Odds ratio"

. collect preview

log(Odds ratio) Odds ratio

X_1 1.3222 * 3.7516 *
(0.5986) (2.2457)

X_2 -1.0648 0.3448
(0.5837) (0.2012)

F_1
On 2.0491 ** 7.7609 **

(0.3343) (2.5944)
Intercept -0.7793 0.4587

(0.4345) (0.1993)



Title and notes

Future update to Stata 17
We will add support for a custom title and notes.

collect title
Specify a custom (common) table title.

Example
. collect title "Logistic regression parameterizations"

. collect preview

Logistic regression parameterizations

log(Odds ratio) Odds ratio

X_1 1.3222 * 3.7516 *
(0.5986) (2.2457)

X_2 -1.0648 0.3448
(0.5837) (0.2012)

F_1
On 2.0491 ** 7.7609 **

(0.3343) (2.5944)
Intercept -0.7793 0.4587

(0.4345) (0.1993)



Title and notes

collect style title
Controls the appearance of table titles.

Example
. collect query title

Table title styles
Collection: default

Common title: Logistic regression parameterizations
Font family:
Font size:

Font color:
Font weight: normal
Font style: normal

Font variant: normal
Font strikeout: normal
Font underline: none

SMCL tag:
LaTeX macro:

Background color:
Foreground color:
Shading pattern:



Title and notes

collect notes
Specify table notes.

Example
. collect notes "** p<=0.01, * p<=0.05"

. collect preview

Logistic regression parameterizations

log(Odds ratio) Odds ratio

X_1 1.3222 * 3.7516 *
(0.5986) (2.2457)

X_2 -1.0648 0.3448
(0.5837) (0.2012)

F_1
On 2.0491 ** 7.7609 **

(0.3343) (2.5944)
Intercept -0.7793 0.4587

(0.4345) (0.1993)

** p<=0.01, * p<=0.05



Title and notes

Example (more notes)
. collect notes "Append notes by default."

. collect notes 0: "Prepend notes by subtle indexing."

. collect notes 2: "** p<=0.01, * p<=0.05 -- replace by index too."

. collect preview

Logistic regression parameterizations

log(Odds ratio) Odds ratio

X_1 1.3222 * 3.7516 *
(0.5986) (2.2457)

X_2 -1.0648 0.3448
(0.5837) (0.2012)

F_1
On 2.0491 ** 7.7609 **

(0.3343) (2.5944)
Intercept -0.7793 0.4587

(0.4345) (0.1993)

Prepend notes by subtle indexing.

** p<=0.01, * p<=0.05 -- replace by index too.
Append notes by default.



Title and notes

collect style notes
Controls the appearance of table notes.

Example
. collect query notes

Table note styles
Collection: default

Note 1: Prepend notes by subtle indexing.
Note 2: ** p<=0.01, * p<=0.05 -- replace by index too.
Note 3: Append notes by default.

Font family:
Font size:

Font color:
Font weight: normal
Font style: normal

Font variant: normal
Font strikeout: normal
Font underline: none

SMCL tag:
LaTeX macro:

Background color:
Foreground color:
Shading pattern:



Estimation tables

Workflow review
I collect get – consume results into a collection
I collect layout – arrange the results in your table
I collect style – control appearance and behaviors
I collect query – show collection properties
I collect label – manipulate and show labels
I collect export – your table to a document



Estimation tables

Challenges

I Too many commands?
I Learn a new language to specify a layout.
I Some behaviors cannot be coded generically in styles.

I Show factor variables without a binder.



Estimation tables

New command: etable
Simple syntax for building estimation tables.

Inspired by community contributed table commands.

Intended to replace estimates table.

Available in a future update to Stata 17.



Estimation tables

etable default behaviors
I Collect from the current estimation results.
I Create a collection named ETable.
I Show dependent variable in the column header.
I Hide equation names.
I Report coefficients.
I Report standard errors with parenthesis.
I Report the number of observations.



Estimation tables

Example (default etable)
. quietly logit

. etable

y1

x1 1.322
(0.599)

x2 -1.065
(0.584)

f1
1 2.049

(0.334)
_cons -0.779

(0.434)
Number of observations 200

. estimates store m1

. etable, estimates(m1)
(same table)



Estimation tables

Use option showstars to label significant coefficients.

Example
. etable, showstars

y1

x1 1.322 *
(0.599)

x2 -1.065
(0.584)

f1
1 2.049 **

(0.334)
_cons -0.779

(0.434)
Number of observations 200



Estimation tables

Example (etable default stars properties)
. collect query stars

Stars style
Collection: ETable

Type: dimension
Results: _r_p
Attach: _r_b

For tags:
Rule 1: .01 **
Rule 2: .05 *
Rule 3:
Rule 4:
Rule 5:

Show note: off
Name of p-value: "p"
Numeric format: "%9.0g"
Note delimiter: ","

Note prefix: ""
Note suffix: ""

Note: ""

stars() option
Change labels and where to attach them.



Estimation tables

etable replaces its collection and consumes from the current
estimation results.

Example
. quietly logit, or

. etable

y1

x1 3.752
(2.246)

x2 0.345
(0.201)

f1
1 7.761

(2.594)
_cons 0.459

(0.199)
Number of observations 200

Use option append to add results to collection ETable.



Estimation tables
Build model comparisons quickly.

Example
. quietly regress y2 x1 x2 i.f1

. estimates store m2

. etable, estimates(m1 m2) showstars

y1 y2

x1 1.322 *
(0.599)

x2 -1.065
(0.584)

f1
1 2.049 **

(0.334)
_cons -0.779

(0.434)
x1 0.691

(0.363)
x2 -0.920 *

(0.355)
f1

1 3.079 **
(0.206)

_cons -0.879 **
(0.276)

Number of observations 200 200



Estimation tables
Use option showeq to show equations.

Example
. etable, estimates(m1 m2) showstars showeq

y1 y2

y1
x1 1.322 *

(0.599)
x2 -1.065

(0.584)
f1

1 2.049 **
(0.334)

_cons -0.779
(0.434)

y2
x1 0.691

(0.363)
x2 -0.920 *

(0.355)
f1

1 3.079 **
(0.206)

_cons -0.879 **
(0.276)

Number of observations 200 200



Estimation tables

Use option replay to apply edits to collection ETable.

Use option eqrecode() to recode equations.

Example
. etable, replay eqrecode(y1 = y y2 = y)

y1 y2

y
x1 1.322 * 0.691

(0.599) (0.363)
x2 -1.065 -0.920 *

(0.584) (0.355)
f1
1 2.049 ** 3.079 **

(0.334) (0.206)
_cons -0.779 -0.879 **

(0.434) (0.276)
Number of observations 200 200



Estimation tables

Example (hide the common equation)
. etable, replay noshoweq

y1 y2

x1 1.322 * 0.691
(0.599) (0.363)

x2 -1.065 -0.920 *
(0.584) (0.355)

f1
1 2.049 ** 3.079 **

(0.334) (0.206)
_cons -0.779 -0.879 **

(0.434) (0.276)
Number of observations 200 200



Estimation tables

Example (change column header to command names)
. etable, replay column(command)

logit regress

x1 1.322 * 0.691
(0.599) (0.363)

x2 -1.065 -0.920 *
(0.584) (0.355)

f1
1 2.049 ** 3.079 **

(0.334) (0.206)
_cons -0.779 -0.879 **

(0.434) (0.276)
Number of observations 200 200



Estimation tables

Example (select your model statistics)
. etable, replay mstat(ll) mstat(N)

logit regress

x1 1.322 * 0.691
(0.599) (0.363)

x2 -1.065 -0.920 *
(0.584) (0.355)

f1
1 2.049 ** 3.079 **

(0.334) (0.206)
_cons -0.779 -0.879 **

(0.434) (0.276)
Log likelihood -109.82 -354.82
Number of observations 200 200



Estimation tables

Example (show variable and factor value labels)
. etable, replay varlabel fvlabel

logit regress

X_1 1.322 * 0.691
(0.599) (0.363)

X_2 -1.065 -0.920 *
(0.584) (0.355)

F_1
On 2.049 ** 3.079 **

(0.334) (0.206)
Intercept -0.779 -0.879 **

(0.434) (0.276)
Log likelihood -109.82 -354.82
Number of observations 200 200



Estimation tables

Example (add title and note)
. etable, replay ///
> title(Model comparison, font(, bold)) ///
> note("** p<=0.01, * p<=0.05", font(, italic))

Model comparison

logit regress

X_1 1.322 * 0.691
(0.599) (0.363)

X_2 -1.065 -0.920 *
(0.584) (0.355)

F_1
On 2.049 ** 3.079 **

(0.334) (0.206)
Intercept -0.779 -0.879 **

(0.434) (0.276)
Log likelihood -109.82 -354.82
Number of observations 200 200

** p<=0.01, * p<=0.05



Estimation tables

Save your style for use as a starting point in a future analysis.

Example
. collect style save my-et-style, replace
(style from ETable saved to file my-et-style.stjson)

set etable_style
You will be able to set a custom default style for etable.



Estimation tables

Rebuild the table using our new style.

Example
. collect clear

. etable, estimates(m1 m2) style(my-et-style)

Model comparison

logit regress

X_1 1.322 * 0.691
(0.599) (0.363)

X_2 -1.065 -0.920 *
(0.584) (0.355)

F_1
On 2.049 ** 3.079 **

(0.334) (0.206)
Intercept -0.779 -0.879 **

(0.434) (0.276)
Log likelihood -109.82 -354.82
Number of observations 200 200

** p<=0.01, * p<=0.05



Estimation tables

Things todo for the Stata 17 update

I Automatic note for stars labels?
I etable support for named scalar expressions?
I Documentation and dialogs for these new features.


